1 Exercise 1

Model Specification uniform/nonuniform DIF in items 11-20

Let Z denote the grouping variable (Z=0,1). The logit of the success probability $\eta_{pi|Z}$ can be specified as follows:

type of DIF	item 1-10	item 11-20
uniform	$\alpha_i\theta_p - \beta_i$	$\alpha_i\theta_p - [\beta_i + Z_p\delta_i^{(\beta)}]$
nonuniform	$\alpha_i\theta_p - \beta_i$	$[\alpha_i + Z_p\delta_i^{(\alpha)}]\theta_p - [\beta_i + Z_p\delta_i^{(\beta)}]$

(1)

SAS Code

The following code can be used to estimate the model:

```
PROC NLMIXED data=dif method=gauss noad technique=newrap qpoints=20;
PARMS b1-b20=0 a1-a20=1 d11-d20=0 e11-e20=0 mu=0 sd=0.5;
alfa=a1*x1+a2*x2+a3*x3+a4*x4+a5*x5+a6*x6+a7*x7+a8*x8+a9*x9+a10*x10
+a11*x11+a12*x12+a13*x13+a14*x14+a15*x15+a16*x16+a17*x17+a18*x18+a19*x19+a20*x20;
beta=b1*x1+b2*x2+b3*x3+b4*x4+b5*x5+b6*x6+b7*x7+b8*x8+b9*x9+b10*x10
+b11*x11+b12*x12+b13*x13+b14*x14+b15*x15+b16*x16+b17*x17+b18*x18+b19*x19+b20*x20;
delta_b=d11*x11+d12*x12+d13*x13+d14*x14+d15*x15+d16*x16+d17*x17+d18*x18+d19*x19+d20*x20;
delta_a=e11*x11+e12*x12+e13*x13+e14*x14+e15*x15+e16*x16+e17*x17+e18*x18+e19*x19+e20*x20;
ex=exp((alfa+Z*delta_a)*theta-beta-Z*delta_b);
p=ex/(1+ex);
MODEL y ~ binary(p);
RANDOM theta ~ normal(Z*mu,(1-Z)+Z*(sd**2)) subject=person;
ESTIMATE 'sd**2' sd**2;
RUN;
```

Model specification RW-DIF for item 12 using dummy- or contrast coding for Z.

For RW-DIF models using dummy- or contrast coding, the logit of the success probability reads as follows:

Coding scheme	Z	item 1-11,13-20	item 12
dummy coding	0	$\alpha_i\theta_p - \beta_i$	$\alpha_i\theta_p - \beta_i$
	1	$\alpha_i\theta_p - \beta_i$	$\alpha_i\theta_p - [\beta_i + \gamma_p]$
effect coding	-1	$\alpha_i\theta_p - \beta_i$	$\alpha_i\theta_p - [\beta_i - \gamma_p]$
	1	$\alpha_i\theta_p - \beta_i$	$\alpha_i\theta_p - [\beta_i + \gamma_p]$

(2)

For the model using dummy coding we specify $p(\theta,\gamma) \sim N([0,0]),[1,0,0]$ for $Z=0$ and $p(\theta,\gamma) \sim N([\mu_{\theta_1},\mu_{\gamma_1}]),[\sigma^2_{\theta_1},\sigma_{\theta\gamma_1},\sigma^2_{\delta_1}]$ for $Z=1$. For the model using contrast coding we specify $p(\theta,\gamma) \sim N([0,0]),[1,\sigma_{\theta\gamma_0},1]$ for $Z=0$ and $p(\theta,\gamma) \sim N([\mu_{\theta_1},\mu_{\gamma_1}]),[\sigma^2_{\theta_1},\sigma_{\theta\gamma_1},\sigma^2_{\delta_1}]$ for $Z=1$.

SAS code for model using dummy coding

The following code can be used to estimate the model:

```
PROC NLMIXED data=dif method=gauss noad technique=newrap qpoints=20;
PARMS b1-b20=0 a1-a20=1 mutheta=0 mugamma=0 sdtheta=.5 sdgamma=.5 cothga=0;
alfa=a1*x1+a2*x2+a3*x3+a4*x4+a5*x5+a6*x6+a7*x7+a8*x8+a9*x9+a10*x10
+a11*x11+a12*x12+a13*x13+a14*x14+a15*x15+a16*x16+a17*x17+a18*x18+a19*x19+a20*x20;
beta=b1*x1+b2*x2+b3*x3+b4*x4+b5*x5+b6*x6+b7*x7+b8*x8+b9*x9+b10*x10
+b11*x11+b12*x12+b13*x13+b14*x14+b15*x15+b16*x16+b17*x17+b18*x18+b19*x19+b20*x20;
ex=exp(alfa*theta-beta-gamma*x12*Z);
p=ex/(1+ex);
MODEL y ~ binary(p);
```

```
RANDOM theta gamma ~ normal([Z*mutheta,Z*mugamma],
[1-Z+Z*sdtheta**2,Z*cothga,Z*sdgamma**2]) subject=person;
RUN;
```

2 Exercise 2

Model Specification uniform/nonuniform DIF in items 6-15

Let Z denote the grouping variable (Z=1,2,3) and let D_1 and D_2 be two dummy variables that can be used to code the information in Z, that is, $D_1 = D_2 = 0$ for $Z = 3$; $D_1 = 1$ and $D_2 = 0$ for $Z = 1$; $D_1 = 0$ and $D_2 = 1$ for $Z = 2$. For models assuming uniform/nonuniform DIF, the logit of the success probability $\eta_{pi|Z}$ can be specified as follows:

type of DIF	item 1-5	item 6-15
uniform	$\alpha_i\theta_p - \beta_i$	$\alpha_i\theta_p - [\beta_i + D_{1p}\delta_{1i}^{(\beta)} + D_{2p}\delta_{2i}^{(\beta)}]$
nonuniform	$\alpha_i\theta_p - \beta_i$	$[\alpha_i + D_{1p}\delta_{1i}^{(\alpha)} + D_{2p}\delta_{2i}^{(\alpha)}]\theta_p - [\beta_i + D_{1p}\delta_{1i}^{(\beta)} + D_{2p}\delta_{2i}^{(\beta)}]$

(3)

SAS code

The following code can be used to simultaneously estimate and test for nonuniform DIF when comparing groups A and C and when comparing groups B and C:

```
PROC NLMIXED data=dif method=gauss noad technique=newrap qpoints=20;
PARMS b1-b15=0 a1-a15=1 d1a6-d1a15=0 d2a6-d2a15=0 d1b6-d1b15=0 d2b6-d2b15=0 mu1-mu2=0
sd1-sd2=0.5;
alfa=a1*x1+a2*x2+a3*x3+a4*x4+a5*x5+a6*x6+a7*x7+a8*x8+a9*x9+a10*x10
+a11*x11+a12*x12+a13*x13+a14*x14+a15*x15;
d1alfa=D1*(d1a6*x6+d1a7*x7+d1a8*x8+d1a9*x9+d1a10*x10
+d1a11*x11+d1a12*x12+d1a13*x13+d1a14*x14+d1a15*x15);
d2alfa=D2*(d2a6*x6+d2a7*x7+d2a8*x8+d2a9*x9+d2a10*x10
+d2a11*x11+d2a12*x12+d2a13*x13+d2a14*x14+d2a15*x15);
beta=b1*x1+b2*x2+b3*x3+b4*x4+b5*x5+b6*x6+b7*x7+b8*x8+b9*x9+b10*x10
+b11*x11+b12*x12+b13*x13+b14*x14+b15*x15;
d1beta=D1*(d1b6*x6+d1b7*x7+d1b8*x8+d1b9*x9+d1b10*x10
+d1b11*x11+d1b12*x12+d1b13*x13+d1b14*x14+d1b15*x15);
d2beta=D2*(d2b6*x6+d2b7*x7+d2b8*x8+d2b9*x9+d2b10*x10
+d2b11*x11+d2b12*x12+d2b13*x13+d2b14*x14+d2b15*x15);
ex=exp((alfa+d1alfa+d2alfa)*theta-beta-d1beta-d2beta);
p=ex/(1+ex);
MODEL y ~ binary(p);
RANDOM theta ~ normal(mu1*D1+mu2*D2,1+D1*sd1**2+D2*sd2**2) subject=person;
ESTIMATE 'sd1**2' sd1**2;
ESTIMATE 'sd2**2' sd2**2;
RUN;
```

Modeling individual differences in DFF using effect coding

Suppose that, in the previous analysis, the last five items (items 11-15), which have a facet F in common, show equally signed uniform DIF when comparing groups A and B to group C and that one would want to model the individual differences in the weight of the facet F. Using effect coding to code the information in Z ($D_1 = D_2 = -1$ for $Z = 3$; $D_1 = 1$ and $D_2 = 0$ for $Z = 1$; $D_1 = 0$ and $D_2 = 1$ for $Z = 2$), the logit of the success probability η_{pi} is specified as follows:

Z	items 1-10	items 11-15
1	$\alpha_i\theta_p - \beta_i$	$\alpha_i\theta_p - \beta_i - \delta_p$
2	$\alpha_i\theta_p - \beta_i$	$\alpha_i\theta_p - \beta_i - \gamma_p$
3	$\alpha_i\theta_p - \beta_i$	$\alpha_i\theta_p - \beta_i + \delta_p + \gamma_p$

(4)

For the random effects we assume a multivariate normal distribution for each of the groups. More specifically, for $Z = 1$ $p(\theta, \delta) \sim \mathrm{N}(\boldsymbol{\mu}_1, \boldsymbol{\Sigma}_1)$ with the identifiability restrictions $\mu_{\theta_1} = 0$ and $\sigma^2_{\theta_1} = 1$; for $Z = 2$ $p(\theta, \gamma) \sim \mathrm{N}(\boldsymbol{\mu}_2, \boldsymbol{\Sigma}_2)$ and for $Z = 3$ $p(\theta, \gamma, \delta) \sim \mathrm{N}(\boldsymbol{\mu}_3, \boldsymbol{\Sigma}_3)$. For Z=2 and Z=3 the parameters of the multivariate distributions can be estimated without further restrictions.

3 Exercise 3

Specification of a partial credit model for trichotomous verbal aggression data that includes behavior-specific dynamic predictors

Let Y_{psbd} and Y_{psbw} $(s = 1\ldots,4; b = 1,\ldots,3)$ denote the observed trichotomous responses (with values 0,1,2) of person p to do and want-items of situation s and behavior b. In order to dynamically model do-responses on the basis of want-responses, we specify the joint distribution $Pr(y_{psbw}, y_{psbd})$ for each situation s and behavior b as a product of two distributions, that is, $Pr(y_{psbw})Pr(y_{psbd}|y_{psbw})$. The distribution of the want-item is modeled with a PCM, that is, we specify:

$$\log\frac{\Pr(Y_{psbw} = 1)}{\Pr(Y_{psbw} = 0)} = \theta_p - \beta_{sbw1}$$

and

$$\log\frac{\Pr(Y_{psbw} = 2)}{\Pr(Y_{psbw} = 1)} = \theta_p - \beta_{sbw2}$$

¿From these logits it follows that

$$\begin{aligned}
&\Pr(Y_{psbw} = 0) = 1/v(\theta_p)\\
&\Pr(Y_{psbw} = 1) = \exp(\theta_p - \beta_{sbw1})/v(\theta_p)\\
&\Pr(Y_{psbw} = 2) = \exp(2\theta_p - \beta_{sbw1} - \beta_{sbw2})/v(\theta_p)
\end{aligned}$$

with $v(\theta_p) = 1 + \exp(\theta_p - \beta_{sbw1}) + \exp(2\theta_p - \beta_{sbw1} - \beta_{sbw2})$.

To model the conditional distribution of the do-item given the want-item behavior-specific dynamic predictors are added to the logits of the PCM. The dynamic predictors are constructed as follows. For each pair (s, b) define a variable X_{psb1} that equals 1 if $Y_{psbw} = 1$ and 0 otherwise and a variable X_{psb2} that equals 1 if $Y_{psbw} = 2$ and 0 otherwise. The behavior-specific dynamic predictors can be obtained summing X_{psb1} and X_{psb2} over situations, that is $X_{p+b1} = \sum_b X_{psb1}$ and $X_{p+b2} = \sum_b X_{psb2}$. The logits for the do-items are specified as follows:

$$\log\frac{\Pr(Y_{psbd} = 1|y_{psbw})}{\Pr(Y_{psbd} = 0|y_{psbw})} = \theta_p - \beta_{sbd1} - \delta_{b1}X_{p+b1} - \delta_{b2}X_{p+b2}$$

and

$$\log\frac{\Pr(Y_{psbd} = 2|y_{psbw})}{\Pr(Y_{psbd} = 1|y_{psbw})} = \theta_p - \beta_{sbd2} - \delta_{b3}X_{p+b1} - \delta_{b4}X_{p+b2}$$

so that

$$\begin{aligned}
&\Pr(Y_{psbd} = 0|y_{psbw}) = 1/v(\theta_p)\\
&\Pr(Y_{psbd} = 1|y_{psbw}) = \exp(\theta_p - \beta_{sbd1} - \delta_{b1}X_{p+b1} - \delta_{b2}X_{p+b2})/v(\theta_p)\\
&\Pr(Y_{psbd} = 2|y_{psbw}) = \exp(2\theta_p - \beta_{sbd1} - \beta_{sbd2} - (\delta_{b1} + \delta_{b3})X_{p+b1} - (\delta_{b2} + \delta_{b4})X_{p+b2})/v(\theta_p)
\end{aligned}$$

with $v(\theta) = 1 + \exp(\theta_p - \beta_{sbd1} - \delta_{b1} X_{p+b1} - \delta_{b2} X_{p+b2}) + \exp(2\theta_p - \beta_{sbd1} - \beta_{sbd2} - (\delta_{b1} + \delta_{b3}) X_{p+b1} - (\delta_{b2} + \delta_{b4}) X_{p+b2})$

SAS Code

```
PROC NLMIXED data=verbal method=gauss technique=newrap noad qpoints=20 maxiter=100
maxfunc=2000;
PARMS b1_1-b1_24=0 b2_1-b2_24=0 dcur1-dcur4=0 dsco1-dsco4=0 dsho1-dsho4=0 sd=.5;
ex1=th-x1*b1_1-x2*b1_2-x3*b1_3-x4*b1_4-x5*b1_5-x6*b1_6-x7*b1_7-x8*b1_8
-x9*b1_9-x10*b1_10-x11*b1_11-x12*b1_12-x13*b1_13-x14*b1_14-x15*b1_15-x16*b1_16
-x17*b1_17-x18*b1_18-x19*b1_19-x20*b1_20-x21*b1_21-x22*b1_22-x23*b1_23-x24*b1_24
-dcur1*dyn1_curse-dcur2*dyn2_curse-dsco1*dyn1_scold-dsco2*dyn2_scold-dsho1*dyn1_shout-dsho2*dyn2_shout;
ex2=th-x1*b2_1-x2*b2_2-x3*b2_3-x4*b2_4-x5*b2_5-x6*b2_6-x7*b2_7-x8*b2_8
-x9*b2_9-x10*b2_10-x11*b2_11-x12*b2_12-x13*b2_13-x14*b2_14-x15*b2_15-x16*b2_16
-x17*b2_17-x18*b2_18-x19*b2_19-x20*b2_20-x21*b2_21-x22*b2_22-x23*b2_23-x24*b2_24
-dcur3*dyn1_curse-dcur4*dyn2_curse-dsco3*dyn1_scold-dsco4*dyn2_scold-dsho3*dyn1_shout-dsho4*dyn2_shout;
denom=(1+exp(ex1)+exp(ex1+ex2));
if (y=0) then p=1/denom;
else if (y=1) then p=exp(ex1)/denom;
else if (y=2) then p=exp(ex1+ex2)/denom;
if (p > 1e-8) then ll=log(p);
else ll=-1e100;
MODEL y   general(ll);
RANDOM th   normal(0,sd**2) subject=pp;
ESTIMATE 'sd**2' sd**2;
RUN;
```

In the above code the parameters dcur1-dcur4, dsco1-dsco4 and dsho1-dsho4 represent the dynamic effects $\delta_{1b}, \ldots, \delta_{4b}$ for cursing, scolding and shouting, respectively.

Results

Estimation of the partial credit model with behavior-specific dynamic effects shows that this model yields a better balance between complexity and fit in terms of AIC and BIC than the original partial credit model: for the partial credit model AIC and BIC equal 12739 and 12923, respectively, whereas for the model with dynamic effects AIC and BIC equal 12384 and 12613, respectively. Table 1 displays the parameter estimates for the dynamic effects along with their standard error and the p-value for the test that the parameter equals zero. The results of the analysis indicate that, for all behaviors, the odds of (actually) displaying a behavior to a moderate extent rather than not at all increase if one also wanted to display the behavior to a moderate extent or to a strong extent. On the other hand, for all behaviors, the odds of actually displaying the behavior to a strong extent rather than to a moderate extent decrease if one wanted to display the behavior to a moderate extent and are not significantly influenced if one wanted to display the behavior to a strong extent.

4 Exercise 4

Model specification for the WDM with behavior-specific dynamic effects and a random dynamic effect for shouting

Let W_{pi1}, W_{pi2}, and W_{pi3} denote behavior-specific dynamic variables for cursing, scolding and shouting that equal 1 for a do-item if the response to the corresponding want-item equals 1. The logit of the probability of a 1-response for the dynamic model that includes behavior-specific fixed effects for cursing and scolding and a behavior-specific random effect for shouting can be expressed

Table 1: Results of partial credit model with behavior-specific dynamic effects

parameter	estimate	standard error	p-value
$\delta_{curse,1}$	-.63	.17	.0003
$\delta_{curse,2}$	-1.03	.21	<.0001
$\delta_{curse,3}$	1.50	.23	<.0001
$\delta_{curse,4}$	.09	.21	.67
$\delta_{scold,1}$	-1.55	.18	<.0001
$\delta_{scold,2}$	-1.52	.22	<.0001
$\delta_{scold,3}$	1.51	.27	<.0001
$\delta_{scold,4}$	.00	.25	.98
$\delta_{shout,1}$	-1.79	.22	<.0001
$\delta_{shout,2}$	-2.19	.26	<.0001
$\delta_{shout,3}$	1.21	.45	.01
$\delta_{shout,4}$	-.27	.40	.51

as:

$$\eta_{pi} = \theta_p - \beta_i + \delta_1 W_{pi1} + \delta_2 W_{pi2} + \gamma_p W_{pi3}.$$

SAS code

1. PROC NLMIXED data=dynam method=gauss noad technique=newrap qpoints=20;
2. PARMS b1-b24=0 d1-d2=0 sdth=0.5 sdga=0.5 cothga=0 muga=0;
3. ex=exp(theta-b1*x1-b2*x2-b3*x3-b4*x4-b5*x5-b6*x6-b7*x7-b8*x8-b9*x9-b10*x10- b11*x11-b12*x12-b13*x13-b14*x14-b15*x15-b16*x16-b17*x17-b18*x18-b19*x19-b20*x20- b21*x21-b22*x22-b23*x23-b24*x24+d1*W1+d2*W2+gamma*W3);
4. p=ex/(1+ex);
5. MODEL y ~ binary(p);
6. RANDOM theta gamma ~ normal([0,muga],[sdth**2,cothga,sdga**2]) subject=pp;
7. predict theta out=empbayes;
8. id gamma;
9. RUN;

Results

Estimation of the model with a random dynamic effect for shouting indicates that this model yields a better balance between complexity and fit in terms of AIC and BIC: For the original fixed-effects model AIC and BIC equal 7975 and 8080, respectively whereas for the model including a random dynamic effect, AIC and BIC equal 7955 and 8068, respectively. Table 2 displays the dynamic effects and the estimated variance-covariance matrix for the distribution $p(\theta, \gamma)$. A comparison with the results of the fixed-effects model (see Table 7.8 in Chapter 7) indicates that the dynamic effects are relatively unaffected by introducing a random effect for shouting. Note that the mean of the γ-dimension (μ_γ) corresponds to the fixed dynamic effect of shouting. Furthermore, the analysis indicates that there is considerable variation on both the θ and the γ dimension: The estimated variances are much larger than their standard errors. Finally, we see that the two estimated dimension are not significantly correlated.

Table 2: Results of Want-Do model with random dynamic effect for shouting

parameter	estimate	standard error	p-value
δ_{curse}	.41	.15	.01
δ_{scold}	1.19	.15	<.0001
μ_{γ}	1.48	.27	<.0001
σ_{θ}	1.27	.07	
σ_{γ}	1.60	.30	
$\sigma_{\theta\gamma}$	.24	.33	.45

5 Exercise 5

Model specification DIF in behavior-specific dynamic effects for males versus females

Let W_{pi1}, W_{pi2}, and W_{pi3} be behavior-specific dynamic variables for cursing, scolding and shouting that equal 1 for a do-item if the response to the corresponding want-item equals 1. Furthermore let Z denote the dummy-coded gender variable (Z=0 for females and Z=1 for males). The logit of the probability of a 1-response for the dynamic model can be expressed as:

$$\eta_{pi|Z} = \theta_p - \beta_i + \sum_{h=1}^{3}(\delta_h + \gamma_h Z_p)W_{pih}$$

SAS code

1. PROC NLMIXED data=dynam method=gauss noad technique=newrap qpoints=20;
2. PARMS b1-b24=0 g1-g3=0 d1-d3=0 sd=0.5;
3. ex=exp(theta-b1*x1-b2*x2-b3*x3-b4*x4-b5*x5-b6*x6-b7*x7-b8*x8-b9*x9-b10*x10- b11*x11-b12*x12-b13*x13-b14*x14-b15*x15-b16*x16-b17*x17-b18*x18-b19*x19-b20*x20- b21*x21-b22*x22-b23*x23-b24*x24 +d1*W1+d2*W2+d3*W3+g1*W1*Z+g2*W2*Z+g3*W3*Z);
4. p=ex/(1+ex);
5. MODEL y ~ binary(p);
6. RANDOM theta ~ normal(0,sd**2) subject=person;
7. RUN;

Results

The results of the analysis (see Table 3) indicate that, for cursing and scolding, the dynamic effects are significantly larger for males than for females (at $\alpha = .05$). In particular, we may calculate that the odds ratio to actually curse rather than not curse given that one wanted to curse versus given that one that did not want to curse is 1.24 for females (i.e. exp(.22)) and 3.32 (i.e. exp(.22+.98)) for males. In the same way the odds ratio to actually scold rather than not scold given that one wanted to scold versus given that one that did not want to scold is 2.74 (i.e. exp(1.01)) for females and 6.42 (i.e. exp(1.01+1.86)) for males.

Model specification behavior-specific dynamic random effects that differ for males and females

Table 3: Results of Want-Do model with behavior-specific dynamic effects that differ for males and females

parameter	estimate	standard error	p-value
δ_{curse}	.22	.16	.16
δ_{scold}	1.01	.16	<.0001
δ_{shout}	1.56	.20	<.0001
γ_{curse}	.98	.24	<.0001
γ_{scold}	.85	.25	.001
γ_{shout}	.48	.28	.08

We will consider a random dynamic effect for scolding, for cursing a similar model can be formulated. For a model with a random dynamic effect for scolding, the logit of the probability of a 1-response reads:

$$\eta_{pi|Z} = \theta_p - \beta_i + \gamma_p W_{pi2} + \delta_1 W_{pi1} + \delta_3 W_{pi3} + \gamma_1 Z_p W_{pi1} + \gamma_3 Z_p W_{pi3}$$

With Z the dummy-coded gender variable. It is assumed that $p(\theta, \gamma) \sim$ MVN($[0, \mu_{\gamma_0}], [\sigma^2_{\theta_0}, \sigma_{\theta\gamma_0}, \sigma^2_{\gamma_0}]$) for females and that $p(\theta, \gamma) \sim$ MVN($[0, \mu_{\gamma_1}], [\sigma^2_{\theta_1}, \sigma_{\theta\gamma_1}, \sigma^2_{\gamma_1}]$) for males.

SAS code

1. PROC NLMIXED data=dynam method=gauss noad technique=newrap qpoints=20;
2. PARMS b1-b24=0 d1=0 d3=0 g1=0 g3=0 muga0=0 muga1=0 sdth0=0.5 sdth1=0.5 sdga0=0.5 sdga1=0.5 cothga0=0 cothga1=0;
3. ex=exp(theta-b1*x1-b2*x2-b3*x3-b4*x4-b5*x5-b6*x6-b7*x7-b8*x8-b9*x9-b10*x10-b11*x11-b12*x12-b13*x13-b14*x14-b15*x15-b16*x16-b17*x17-b18*x18 -b19*x19-b20*x20-b21*x21-b22*x22-b23*x23-b24*x24 +d1*W1+d3*W3+g1*W1*Z+g3*W3*Z+gamma*W2);
4. p=ex/(1+ex);
5. MODEL y ∼ binary(p);
6. RANDOM theta gamma∼ normal([0,(1-Z)*muga0+Z*muga1], [(1-Z)*sdth0**2+Z*sdth1**2, (1-Z)*cothga0**2+Z*cothga1**2, (1-Z)*sdga0**2+Z*sdga1**2]) subject=person;
7. RUN;

Results

Table 4 shows the estimates of the fixed dynamic effects and the variance-covariance matrix of $p(\theta, \gamma)$ for males and females. The results of the present analysis lead to very similar conclusions as the fixed effects analysis, but they also indicate that the dynamic effect of scolding considerably differs among females.

Table 4: Results of a Want-Do model with a random dynamic effect for scolding

parameter	estimate	standard error	p-value
δ_{curse}	.22	.16	.16
δ_{shout}	1.56	.20	<.0001
γ_{curse}	.98	.24	<.0001
γ_{shout}	.51	.28	.07
μ_{γ_0}	.98	.21	<.0001
μ_{γ_1}	2.00	.30	<.0001
σ_{θ_0}	1.28	.08	
σ_{θ_1}	1.09	.13	
σ_{γ_0}	1.37	.32	
σ_{γ_1}	.58	.58	
$\sigma_{\theta\gamma_0}$	.76	.36	.03
$\sigma_{\theta\gamma_1}$	-.18	.36	.62